

## Birats: a bivariate normal hierarchical model


We return to the Rats example, and illustrate the use of a multivariate Normal (MVN) population distribution for the regression coefficients of the growth curve for each rat. This is the model adopted by Gelfand et al (1990) for these data, and assumes *a priori* that the intercept and slope parameters for each rat are correlated. For example, positive correlation would imply that initially heavy rats (high intercept) tend to gain weight more rapidly (steeper slope) than lighter rats. The model is as follows

$$\begin{aligned} Y_{ij} &\sim \text{Normal}(\mu_{ij}, \tau_c) \\ \mu_{ij} &= \beta_{1i} + \beta_{2i} x_j \\ \beta_i &\sim \text{MVN}(\mu_\beta, \Omega) \end{aligned}$$

where  $Y_{ij}$  is the weight of the  $i$ th rat measured at age  $x_j$ , and  $\beta_i$  denotes the vector  $(\beta_{1i}, \beta_{2i})$ . We assume 'non-informative' independent univariate Normal priors for the separate components  $\mu_{\beta_1}$  and  $\mu_{\beta_2}$ . A Wishart( $R, \rho$ ) prior was specified for  $\Omega$ , the population precision matrix of the regression coefficients. To represent vague prior knowledge, we chose the the degrees of freedom  $\rho$  for this distribution to be as small as possible (i.e. 2, the rank of  $\Omega$ ). The scale matrix was specified as

$$R = \begin{vmatrix} 200, & 0 \\ 0, & 0.2 \end{vmatrix}$$

This represents our prior guess at the order of magnitude of the *covariance* matrix  $\Omega^{-1}$  for  $\beta_i$  (see Classic BUGS manual (version 0.5) section on Multivariate normal models), and is equivalent to the prior specification used by Gelfand et al. Finally, a non-informative Gamma(0.001, 0.001) prior was assumed for the measurement precision  $\tau_c$ .



```

model
{
  for( i in 1 : N ) {
    beta[i, 1 : 2] ~ dmnorm(mu.beta[], R[ , ])
    for( j in 1 : T ) {
      Y[i, j] ~ dnorm(mu[i, j], tauC)
      mu[i, j] <- beta[i, 1] + beta[i, 2] * x[j]
    }
  }

  mu.beta[1 : 2] ~ dmnorm(mean[], prec[ , ])
  R[1 : 2 , 1 : 2] ~ dwish(Omega[ , ], 2)
  tauC ~ dgamma(0.001, 0.001)
  sigma <- 1 / sqrt(tauC)
}

```

[Data](#) ( click to open )

[Inits](#) ( click to open )

**Results**

A 1000 update burn in followed by a further 10000 updates gave the parameter estimates

	<b>mean</b>	<b>sd</b>	<b>MC_error</b>	<b>val2.5pc</b>	<b>median</b>	<b>val97.5pc</b>	<b>start</b>	<b>sample</b>
mu.beta[1]	106.6	2.35	0.0335	101.8	106.6	111.2	1001	10000
mu.beta[2]	6.185	0.1062	0.001351	5.981	6.185	6.397	1001	10000
sigma	6.136	0.4781	0.009095	5.283	6.1	7.137	1001	10000